# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: nut543 <kfs1@online.no>

pkgname='freedroidrpg'
pkgdesc="Isometric 3D role playing game taking place in the future, on Earth"
pkgver=1.0
pkgrel=1
arch=('loong64' 'x86_64')
url="https://gitlab.com/freedroid/freedroid-src"
license=('GPL')
depends=('sdl_mixer' 'sdl_image' 'sdl_gfx' 'libogg' 'libvorbis' 'libgl' 'glu' 'glew' 'hicolor-icon-theme')
makedepends=('mesa' 'python')
optdepends=('python' 'espeak-ng-espeak')
source=(https://ftp.osuosl.org/pub/freedroid/freedroidRPG-1.0/freedroidRPG-$pkgver.tar.gz)
sha256sums=('7995b70b594248ea14d1b4ef595397a60183031c99163b01c1a8a70ccef3bbcf')

build() {
  cd $pkgname-${pkgver^^}
  export CPPFLAGS="$CPPFLAGS -fcommon -fPIE"
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-${pkgver^^}
  make DESTDIR="${pkgdir}" install
}
